

Lecture 6: Designing Sequential Logic


Lpset 5 out now, due Thursday

Lab 02 Part 2 due today/tomorrow!

Lab 03 out, due in a week!



Previously on 6.111

- We've started to discuss how to design things in stages:
 - Split operations down into collections of combinational logic isolated by register/flip-flops,
 - Our designs become functions of time



D-Register Timing 1

 =undetermined state



IMPORTANT:

t_{PD} : maximum propagation delay, @posedge CLK $D \rightarrow Q$

Maximum time it takes for Q to change after rising edge of CLK

t_{CD} : minimum contamination delay, @posedge CLK $D \rightarrow Q$

Minimum time it takes for Q to start to change after rising edge of CLK

t_{SETUP} : setup time

How long D must be stable **before** the rising edge of CLK

t_{HOLD} : hold time

How long D must be stable **after** the rising edge of CLK


**New timing attributes
for registers**

Huh?

- In Lab 3:


```
1 module synchronize #(parameter NSYNC = 3) // number of sync flops. must be >= 2
2     (input clk_in, val_in,
3      output logic val_out);
4     logic [NSYNC-2:0] sync;
5
6     always_ff @ (posedge clk_in) begin
7         {val_out, sync} <= {sync[NSYNC-2:0], val_in};
8     end
9 endmodule
```

- Basically builds this:




What if...?

- ...we need to interface with outside equipment:
 - Other systems (on different clocks or from combinational logic?)
 - Human-based inputs (buttons)




Can't guarantee setup and hold times will be met!


Example: Asynchronous Inputs in Sequential Systems




When an asynchronous signal causes a setup/hold violation...



Transition is missed on first clock cycle, but caught on next clock cycle.



Transition is caught on first clock cycle.



Output is metastable for an indeterminate amount of time.

Q: Which cases are problematic?

Metastability

- D-registers have issues with all that feedback and stuff going on.
Can go **metastable**
- Metastability is where the system hovers between Logic High and Logic Low in an unpredictable way




Figure 2. Effects of Violating t_{SU} & t_H Requirements





Figure 1. Metastability Timing Parameters



Metastability in Altera (®) Devices
Altera Application Note 42 (1999)

t_{CO} = "min time from clock to output"
....think of it as t_{pd} here (not exactly the same,

Handling Metastability

- Can't globally prevent metastability, but can isolate it!
- Stringing several registers together can isolate any freakouts!





Figure 8. Two-flip-flop synchronization circuit.




“Metastability and Synchronizers: A Tutorial”

Ran Ginosar, Technion Israel Institute of Technology

Handling Metastability

- Completely preventing metastability turns out to be an impossible problem
- High gain of digital devices makes it likely that metastable conditions will resolve themselves quickly
- Solution to metastability: allow time for signals to stabilize





How many registers are necessary in 6.111?

- Depends on many design parameters (clock speed, device speeds, ...)
- In 6.111, a **pair of synchronization** registers is sufficient
- And for simple designs...with low t_{pd} you may not even need anything


Handling Metastability

- Don't break off an asynchronous input until it has gone through some registers
- Basically: Ensure that external signals feed exactly one flip-flop




Mean Time Between Failures

Figure 1. Metastability Timing Parameters




Set by user (how much extra time do you provide per cycle for metastability to dissipate)


Figure 5. Metastability Characteristics of Altera Devices



D Register Timing 2



— =determined state
 =undetermined state




**Two Requirements/
Conclusions:**

$$t_{PD,reg1} + t_{PD,logic} + t_{SETUP,reg2} \leq t_{CLK}$$


$$t_{CD,reg1} + t_{CD,logic} \geq t_{HOLD,reg2}$$

D Register Timing 2



— =determined state

▨ =undetermined state




**Two Requirements/
Conclusions:**

$$t_{PD,reg1} + t_{met} + t_{PD,logic} + t_{SETUP,reg2} \leq t_{CLK}$$

$$t_{CD,reg1} + t_{CD,logic} \geq t_{HOLD,reg2}$$

Clock Domain Crossing

- For example:
 - Data gets sent in at 25 MHz from one device (running on its own clock)
 - Your system runs at 50 MHz



```
1 //xfer_pipe can be >2 bits wide (2 is usually fine...3 better)
2 always_ff @(posedge new_clock)
3   { new_val, xfer_pipe } <= { xfer_pipe, i_val };
```


- This only works when original clock domain frequency is \leq to new clock domain frequency

FSM Design

...What is a structured way to go about designing state machines?

Design Example: Level-to-Pulse

- A **level-to-pulse converter** produces a single-cycle pulse each time its input goes high.
- It's a **synchronous** rising-edge detector.
- Sample uses:
 - Buttons and switches pressed by humans for arbitrary periods of time
 - Single-cycle enable signals for counters



Leve-to-Pulse


- One simple solution (~from Lab 2)
- One bit positive discrete time positive

```
1  module simple_soln(input clk_in, input l_in, output logic p_out);  
2      logic old_l_in; //remember previous value!  
3      assign p_out = l_in & ~old_l_in; //high and prev low  
4      always_ff @(posedge clk)  
5          old_l_in <= l_in; //remember it!  
6  endmodule  
7
```

- Let's try to formalize this a bit more

Finite State Machines

- Finite State Machines (FSMs) are a useful abstraction for **sequential circuits** with centralized “**states**” of operation
- At each clock edge, combinational logic computes **outputs** and **next state** as a function of **inputs** and **present state**



Level-to-Pulse

- **State:** how/what stores past information?
- **Output Logic:** How does state and input influence output
- **State Transition Logic:** Logic dictating next state
- **State Transition:** Actual updating of state

State

```
1 module simple_soln(input clk_in, input l_in, output logic p_out);  
2   logic old_l_in; //remember previous value!  
3   assign p_out = l_in & ~old_l_in; //high and prev low  
4   always_ff @(posedge clk)  
5     old_l_in <= l_in; //remember it!  
6 endmodule  
7
```

State Transition
and


State Transition Logic

Output Logic


Let's Formalize it: Two Types of FSMs

Moore and **Mealy** FSMs : different output generation

- **Moore FSM:**



- **Mealy FSM:**



Moore



- Edward F. Moore
- 1925-2003
- Virginia Tech
- Worked with Claude Shannon
- Not same Moore as Moore's Law...that was Gordon Moore from Intel


Mealy



- George H. Mealy
- 1927-2010
- Harvard, Bell Labs


Design Example: Level-to-Pulse

- A **level-to-pulse converter** produces a single-cycle pulse each time its input goes high.
- It's a synchronous rising-edge detector.
- Sample uses:
 - Buttons and switches pressed by humans for arbitrary periods of time
 - Single-cycle enable signals for counters




Step 1: State Transition Diagram


- Block diagram of desired system:



- State transition diagram** is a useful FSM representation and design aid:



Valid State Transition Diagrams



- Arcs leaving a state are **mutually exclusive**, i.e., for any combination input values there's at most one applicable arc
- Arcs leaving a state are **collectively exhaustive**, i.e., for any combination of input values there's at least one applicable arc**
- So for each state: for any combination of input values there's exactly one applicable arc (**no ambiguity**)
- Often a starting state is specified
- Each state specifies values for all outputs (in the case of Moore)

Choosing State Representation

Choice #1: **binary encoding**


For N states, use $\text{ceil}(\log_2 N)$ bits to encode the state with each state represented by a unique combination of the bits. Tradeoffs: most efficient use of state registers, but requires more complicated combinational logic to detect when in a particular state.

Choice #2: **“one-hot” encoding**

For N states, use N bits to encode the state where the bit corresponding to the current state is 1, all the others 0. Tradeoffs: more state registers, but often much less combinational logic since state decoding is trivial.


Step 2: Logic Derivation

Transition diagram is readily converted to a state transition table (just a truth table)




Current State		In	Next State		Out
S_1	S_0	L	S_1^+	S_0^+	P
0	0	0	0	0	0
0	0	1	0	1	0
0	1	0	0	0	1
0	1	1	1	1	1
1	1	0	0	0	0
1	1	1	1	1	0


- Combinational logic **could** be derived using Karnaugh maps, but we'll let an HDL do that for us



Moore Level-to-Pulse Converter



Moore FSM circuit implementation of level-to-pulse converter:



Moore Level-to-Pulse Converter (SystemVerilog)

- An example of a **very explicit** Moore FSM implementation of the level-to-pulse converter:


```
1  module moore_fsm(input clk_in, input l_in, output logic p_out);
2      parameter LOW_WAITING = 2'b0;    //define your states as...
3      parameter EDGE_DETECTED = 2'b01; //parameters for easy...
4      parameter HIGH_WAITING = 2'b10;  //reading!
5
6
7      logic [1:0] state;    //contain state!
8      logic [1:0] next_state; //hold next state!
9
10     //Output Logic:
11     always_comb begin
12         case(state)
13             LOW_WAITING: p_out = 1'b0; //output based only on...
14             EDGE_DETECTED: p_out = 1'b1; //current state! This is...
15             HIGH_WAITING: p_out = 1'b0; //a characteristic of Moore FSM
16             default:      p_out = 1'b0;
17         endcase
18     end
19
20     //State Transition Logic (Combinational):
21     always_comb begin
22         case(state) //Also consider explicit if/elses
23             LOW_WAITING: next_state = l_in?EDGE_DETECTED:LOW_WAITING;
24             EDGE_DETECTED: next_state = l_in?HIGH_WAITING:EDGE_DETECTED;
25             HIGH_WAITING: next_state = l_in?HIGH_WAITING:LOW_WAITING;
26             default:      next_state = LOW_WAITING;
27         endcase
28     end
29
30     //State Transition
31     always_ff @(posedge clk_in) begin
32         //consider adding a reset here as well!
33         state <= next_state; //state becomes calculated next_state
34     end
35 endmodule
```

Moore Level-to-Pulse Converter (SystemVerilog)



- Merging **State Transition Logic** and **State Transition** into one block
- Some people like this more (me)

```
1  module moore_fsm(input clk_in, input l_in, output logic p_out);
2      parameter LOW_WAITING = 2'b0;
3      parameter EDGE_DETECTED = 2'b01;
4      parameter HIGH_WAITING = 2'b10;
5
6      logic [1:0] state;
7
8      //Output Logic:
9      always_comb begin
10         case(state)
11             LOW_WAITING:    p_out = 1'b0;
12             EDGE_DETECTED:  p_out = 1'b1;
13             HIGH_WAITING:   p_out = 1'b0;
14             default:        p_out = 1'b0; //default
15         endcase
16     end
17
18     //State Transition and Logic:
19     always_ff @(posedge clk_in) begin
20         //consider adding a reset here as well!
21         case(state)
22             LOW_WAITING:    state <= l_in?EDGE_DETECTED:LOW_WAITING;
23             EDGE_DETECTED:  state <= l_in?HIGH_WAITING:EDGE_DETECTED;
24             HIGH_WAITING:   state <= l_in?HIGH_WAITING:LOW_WAITING;
25             default:        state <= LOW_WAITING;
26         endcase
27     end
28 endmodule
```

Design of a Mealy Level-to-Pulse




- Since outputs are determined by state *and* inputs, Mealy FSMs may need fewer states than Moore FSM implementations




Output transitions immediately.
State transitions at the clock edge.

Mealy Level-to-Pulse Converter



Pres. State	In	Next State	Out
S	L	S^+	P
0	0	0	0
0	1	1	1
1	1	1	0
1	0	0	0

Mealy FSM circuit implementation of level-to-pulse converter:



- FSM's state simply remembers the previous value of L
- Circuit benefits from the Mealy FSM's implicit single-cycle assertion of outputs during state transitions

Mealy Level-to-Pulse Converter (SystemVerilog)

- An example of a **very explicit** Mealy FSM implementation of the level-to-pulse converter:

```
1  module mealy_fsm(input clk_in, input l_in, output logic p_out);
2      parameter LOW_WAITING = 1'b0; //define states but notice...
3      parameter HIGH_WAITING = 1'b1; //fewer needed...Mealy usually...
4                                          //though not always, is like that
5
6
7      logic state; //state (smaller than before...only two states to rep)
8      logic next_state;
9
10     //Output Logic:
11     always_comb begin
12         case(state) //outputs are based on state AND inputs!
13             LOW_WAITING: p_out = l_in?1'b1:1'b0;
14             HIGH_WAITING: p_out = 1'b0;
15             default: p_out = 1'b0; //default
16         endcase
17     end
18
19     //State Transition Logic:
20     always_comb begin
21         case(state)
22             LOW_WAITING: next_state = l_in?HIGH_WAITING:LOW_WAITING;
23             HIGH_WAITING: next_state = l_in?HIGH_WAITING:LOW_WAITING;
24             default: next_state = LOW_WAITING;
25         endcase
26     end
27     //State Transition
28     always_ff @(posedge clk_in) begin
29         //consider adding a reset here as well (same goes for any...
30         //clocked logic block)
31         state <= next_state;
32     end
33 endmodule
```


Mealy Level-to-Pulse Converter (SystemVerilog)


- Merging **State Transition Logic** and **State Transition** into one block

```
1
2 module mealy_fsm(input clk_in, input l_in, output logic p_out);
3     parameter LOW_WAITING = 1'b0;
4     parameter HIGH_WAITING = 1'b1;
5
6     logic state;
7
8     //Output Logic:
9     always_comb begin
10         case(state)
11             LOW_WAITING:    p_out = l_in?1'b1:1'b0;
12             HIGH_WAITING:   p_out = 1'b0;
13             default:        p_out = 1'b0; //default
14         endcase
15     end
16
17     //State Transition and Transition Logic!
18     always_ff @(posedge clk_in) begin
19         //consider adding a reset here as well!
20         case(state)
21             LOW_WAITING:    state <= l_in?HIGH_WAITING:LOW_WAITING;
22             HIGH_WAITING:   state <= l_in?HIGH_WAITING:LOW_WAITING;
23             default:        state <= LOW_WAITING;
24         endcase
25     end
26 endmodule
```


Moore/Mealy Trade-Offs

- How are they different?
 - Moore: **outputs = f(state)** only
 - Mealy **outputs = f(state *and* input)**
 - Mealy outputs generally occur one cycle earlier than a Moore:

Moore: delayed assertion of P




Mealy: immediate assertion of P



- Compared to a Moore FSM, a Mealy FSM ***might***...
 - Be more difficult to conceptualize and design (both at circuit level and in HDL)
 - Have fewer states
 - Be expressed using fewer lines of Verilog

Moore/Mealy Trade-Offs

- Moore:
 - Usually more states
 - Each state has a particular output
- Mealy:
 - Fewer states, outputs are specified on edges of diagram
 - Potential Dangers:



*Possible cyclic logic paths
Combinatorial logic driving itself
asynchronously through really
hard-to-debug pathways!*

FSM Example

GOAL:

- Build an electronic combination lock with a reset button, two number buttons (0 and 1), and an unlock output signal. The combination will **always be 01011**.
- Use a sliding window of the last five entries




STEPS:


1. Design lock FSM (block diagram, state transitions)
2. Write SystemVerilog module(s) for FSM

Step 1A: Block Diagram

lock




Step 1B: State transition diagram



Step 2: Write Verilog

```
1  module lock(input clk,rst_in,b0_in,b1_in,  
2             output logic unlock_out);  
3  
4      // implement state transition diagram  
5      logic [2:0] state,next_state;  
6      always_comb begin  
7          // combinational logic!  
8          next_state = ???;  
9      end  
10     always_ff @(posedge clk_in) state <= next_state;  
11  
12     // generate output  
13     assign out = ???;  
14  
15 endmodule
```


Step 2B: state transition diagram



```

5  parameter S_RESET = 0;  parameter S_0 = 1; // state assigned
6  parameter S_01 = 2;    parameter S_010 = 3;
7  parameter S_0101 = 4;  parameter S_01011 = 5;
8
9  logic [2:0] state, next_state; //(both 3 bits wide)
10
11 always_comb begin // implement state transition diagram
12     if (rst_in) next_state = S_RESET;
13     else case (state)
14         S_RESET: next_state = b0_in ? S_0      : b1_in ? S_RESET : state;
15         S_0:      next_state = b0_in ? S_0      : b1_in ? S_01     : state;
16         S_01:     next_state = b0_in ? S_010    : b1_in ? S_RESET  : state;
17         S_010:    next_state = b0_in ? S_0      : b1_in ? S_0101   : state;
18         S_0101:   next_state = b0_in ? S_010    : b1_in ? S_01011  : state;
19         S_01011:  next_state = b0_in ? S_0      : b1_in ? S_RESET  : state;
20         default:  next_state = S_RESET;        // handle unused states
21     endcase
22 end
23 always_ff @(posedge clk) state <= next_state;
24

```


Step 2C: generate output


```
// it's a Moore machine! Output only depends on current state
```

```
25  assign unlock_out = (state == S_01011);    // assign output: Moore machine
```

Step 2: final Verilog implementation

```
1
2 module lock(input clk_in,rst_in,b0_in,b1_in,
3             output logic unlock_out);
4
5     parameter S_RESET = 0;   parameter S_0 = 1; // state assignments
6     parameter S_01 = 2;      parameter S_010 = 3;
7     parameter S_0101 = 4;    parameter S_01011 = 5;
8
9     logic [2:0] state, next_state; //(both 3 bits wide)
10
11     always_comb begin // implement state transition diagram
12         if (rst_in) next_state = S_RESET;
13         else case (state)
14             S_RESET: next_state = b0_in ? S_0 : b1_in ? S_RESET : state;
15             S_0:      next_state = b0_in ? S_0 : b1_in ? S_01 : state;
16             S_01:     next_state = b0_in ? S_010 : b1_in ? S_RESET : state;
17             S_010:    next_state = b0_in ? S_0 : b1_in ? S_0101 : state;
18             S_0101:   next_state = b0_in ? S_010 : b1_in ? S_01011 : state;
19             S_01011:  next_state = b0_in ? S_0 : b1_in ? S_RESET : state;
20             default:  next_state = S_RESET; // handle unused states
21         endcase
22     end
23     always_ff @(posedge clk) state <= next_state;
24
25     assign unlock_out = (state == S_01011); // assign output: Moore machine
26 endmodule
27
```

Real FSM Security System



The 6.111 Vending Machine

(example from circa 2000...slightly updated)

- Lab assistants demand a new soda machine for the 6.111 lab. You design the FSM controller.
- **All selections are \$0.30.**
- The machine makes change. (Dimes and nickels only.)
- Inputs: limit 1 per clock
 - Q - quarter inserted
 - D - dime inserted
 - N - nickel inserted
- Outputs: limit 1 per clock
 - DC - dispense can
 - DD - dispense dime
 - DN - dispense nickel



What States are in the System?

- A starting (idle) state:



- A state for each possible amount of money captured:



- What's the maximum amount of money captured before purchase?
25 cents (just shy of a purchase) + one quarter (largest coin)




- States to dispense change (one per coin dispensed):




A Moore Vender

Here's a first cut at the state transition diagram.




State Reduction




17 states
5 state bits

15 states
4 state bits



Verilog for the Moore Vender



- **State register**
(sequential always block)
- **Next-state combinational logic**
(comb. always block with case)
- **Output combinational logic block**
(comb. always block *or* assign statements)

So triggered on posedge clock

```
1 module mooreVender (  
2     input N, D, Q, clk, reset,  
3     output DC, DN, DD,  
4     output logic [3:0] state);  
5  
6     logic next;  
7
```

States defined with **parameter** keyword

```
10  
11     parameter IDLE = 0;  
12     parameter GOT_5c = 1;  
13     parameter GOT_10c = 2;  
14     parameter GOT_15c = 3;  
15     parameter GOT_20c = 4;  
16     parameter GOT_25c = 5;  
17     parameter GOT_30c = 6;  
18     parameter GOT_35c = 7;  
19     parameter GOT_40c = 8;  
20     parameter GOT_45c = 9;  
21     parameter GOT_50c = 10;  
22     parameter RETURN_20c = 11;  
23     parameter RETURN_15c = 12;  
24     parameter RETURN_10c = 13;  
25     parameter RETURN_5c = 14;
```

State register defined with **sequential always** block (**always_ff**)

```
26  
27  
28     always_ff @(posedge clk or negedge reset) begin  
29         if (!reset) state <= IDLE;  
30         else state <= next;  
31     end  
32
```


Next-state logic within a **combinational always** block



Verilog for the Moore Vender

Combinational output assignment

```
1
2 always_comb (state or N or D or Q) begin
3     case (state)
4         IDLE:      if (Q) next = GOT_25c;
5                     else if (D) next = GOT_10c;
6                     else if (N) next = GOT_5c;
7                     else next = IDLE;
8         GOT_5c:    if (Q) next = GOT_30c;
9                     else if (D) next = GOT_15c;
10                    else if (N) next = GOT_10c;
11                    else next = GOT_5c;
12         GOT_10c:   if (Q) next = GOT_35c;
13                    else if (D) next = GOT_20c;
14                    else if (N) next = GOT_15c;
15                    else next = GOT_10c;
16         GOT_15c:   if (Q) next = GOT_40c;
17                    else if (D) next = GOT_25c;
18                    else if (N) next = GOT_20c;
19                    else next = GOT_15c;
20         GOT_20c:   if (Q) next = GOT_45c;
21                    else if (D) next = GOT_30c;
22                    else if (N) next = GOT_25c;
23                    else next = GOT_20c;
24         GOT_25c:   if (Q) next = GOT_50c;
25                    else if (D) next = GOT_35c;
26                    else if (N) next = GOT_30c;
27                    else next = GOT_25c;
28         GOT_30c:   next = IDLE;
29         GOT_35c:   next = RETURN_5c;
30         GOT_40c:   next = RETURN_10c;
31         GOT_45c:   next = RETURN_15c;
32         GOT_50c:   next = RETURN_20c;
33         RETURN_20c: next = RETURN_10c;
34         RETURN_15c: next = RETURN_5c;
35         RETURN_10c: next = IDLE;
36         RETURN_5c:  next = IDLE;
37         default:   next = IDLE;
38     endcase
39 end
```

```
1 assign DC = (state == GOT_30c || state == GOT_35c ||
2             state == GOT_40c || state == GOT_45c ||
3             state == GOT_50c);
4 assign DN = (state == RETURN_5c);
5 assign DD = (state == RETURN_20c || state == RETURN_15c ||
6             state == RETURN_10c);
7 endmodule
```


Simulation of Moore Vender




FSM Output Glitching

- FSM state bits may not transition at precisely the same time
- Combinational logic for outputs may contain hazards
- Result: your FSM outputs may glitch!


during this state transition...



...the state registers may transition like this...



*...causing the DC output to **glitch** like this!*




```
1 assign DC = (state == GOT_30c || state == GOT_35c ||  
2 state == GOT_40c || state == GOT_45c ||  
3 state == GOT_50c);
```


If the soda dispenser is glitch-sensitive, your customers can get a 20-cent soda!

One way to fix Glitches:

- Don't have to have state 3 (3'b011) go into state 4 (3'b100). Use different state naming/use different numbers!!! *A rose by any other name would smell as sweet*
- Perhaps a Gray code (??):
 - Count up like: 000, 001, 011, 010, 110, 111, 101, 100, ...
 - Have the really important/glitch-sensitive states only require transitions of one bit
- One-hot encoding:



Another Solution: Registered FSM Outputs are Glitch-Free



- Move output generation into the sequential always block
- Calculate outputs based on next state
- Delays outputs by one clock cycle. Problematic in some application.


```
1  always_ff @(posedge clk or negedge reset) begin
2      if (!reset) state <= IDLE;
3      else if (clk) state <= next;
4
5      DC <= (next == GOT_30c || next == GOT_35c ||
6            next == GOT_40c || next == GOT_45c ||
7            next == GOT_50c);
8      DN <= (next == RETURN_5c);
9      DD <= (next == RETURN_20c || next == RETURN_15c ||
10           next == RETURN_10c);
11 end
```

*Note this is inside an edged always with non-blocking assigns!
This will synthesize to registered outputs!*

Let's Do a Better Lock

GOAL:

- Build an electronic combination lock with a reset button, two number buttons (0 and 1), and an unlock output signal. The combination is 5 bits **specified by the user!**.
- After each attempt, reset must be pushed to try again



By departing a bit from a rigid FSM structure we can maybe express ourselves more effectively!!

Implementation 1

- Rather verbose Moore FSM
- Output is Moore-ish (but also additional information)

```
2 module lock_fsm (input clk_in, rst_in, b0_in, b1_in, [4:0] code_in,
3                 output logic unlock_out);
4     parameter IDLE = 0;
5     parameter ENTRY_1 = 1;
6     parameter ENTRY_2 = 2;
7     parameter ENTRY_3 = 3;
8     parameter ENTRY_4 = 4;
9     parameter ENTRY_5 = 5;
10
11     logic [2:0] state; //3 bits 8 states!
12     logic [4:0] vals_entered;
13
14     //Output Logic:
15     assign unlock_out = (state==ENTRY_5)&&(vals_entered == code_in);
16
17     //State Transition and Transition Logic!
18     always_ff @(posedge clk_in) begin
19         if (rst_in)begin
20             state <= IDLE;
21             vals_entered <= 5'b0;
22         end else begin
23             if (state != ENTRY_5 && (b0_in||b1_in))begin
24                 if (b0_in)
25                     vals_entered <= {vals_entered[3:0],0};
26                 else
27                     vals_entered <= {vals_entered[3:0],1};
28             end
29             case(state)
30                 IDLE:
31                     state <= (b0_in||b1_in)?ENTRY_1;
32                 ENTRY_1:
33                     state <= (b0_in||b1_in)?ENTRY_2;
34                 ENTRY_2:
35                     state <= (b0_in||b1_in)?ENTRY_3;
36                 ENTRY_3:
37                     state <= (b0_in||b1_in)?ENTRY_4;
38                 ENTRY_4:
39                     state <= (b0_in||b1_in)?ENTRY_5;
40                 ENTRY_5:
41                     state <= ENTRY_5; //stays in place until reset
42                 default:
43                     state <= IDLE;
44             endcase
45         end
46     end
47 endmodule
```


Implementation 1

- Very similar to previous implementation, but took advantage of ordering of states

```
1  module lock_fsm (input clk_in, rst_in, b0_in, b1_in, [4:0] code_in,  
2                      output logic unlock_out);  
3  
4      parameter ENTRY_0 = 0; parameter ENTRY_1 = 1;  
5      parameter ENTRY_2 = 2; parameter ENTRY_3 = 3;  
6      parameter ENTRY_4 = 4; parameter ENTRY_5 = 5;  
7  
8      logic [2:0] state; //3 bits 8 states!  
9      logic [4:0] vals_entered;  
10  
11     //Output Logic:  
12     assign unlock_out = (state==ENTRY_5)&&(vals_entered == code_in);  
13  
14     //State Transition and Transition Logic!  
15     always_ff @(posedge clk_in) begin  
16         if (rst_in)begin  
17             state <= ENTRY_0;  
18             vals_entered <= 5'b0;  
19         end else begin  
20             if (state < ENTRY_5 && (b0_in||b1_in))begin  
21                 state <= state +1;  
22                 vals_entered <= {vals_entered[3:0],b1_in};  
23             end  
24         end  
25     end  
26 endmodule  
27
```

*Cute but sneaky way to add in
which button is getting pushed*

Implementation 2

- Use fewer explicit states, but maybe need to remember stuff somehow
- Bring in additional variables...in theoretical sense those are also states

A rose by any other name...

```
1
2 module lock_fsm (input clk_in, rst_in, b0_in, b1_in, [4:0] code_in,
3                 output logic unlock_out);
4     parameter IDLE = 0;
5     parameter ENTRY = 1;
6
7     logic [3:0] entered_count;
8     logic state; //1 bits 2 states!
9     logic [4:0] vals_entered;
10
11 //Output Logic:
12 assign unlock_out = (state==ENTRY)&&(entered_count==5)&&(vals_entered == code_in);
13
14 //State Transition and Transition Logic!
15 always_ff @(posedge clk_in) begin
16     if (rst_in)begin
17         state <= IDLE;
18         vals_entered <= 5'b0;
19         entered_count <= 4'b0;
20     end else if (entered_count < 5 && (b0_in||b1_in))begin
21         state <= ENTRY;
22         vals_entered <= {vals_entered[3:0],b1_in};
23         entered_count <= entered_count + 1;
24     end
25 end
26 endmodule
27
```

Implementation 2

- If I'm worried about glitching...
- Move unlock_out assignment to within sequential block
- Becomes registered, far less likely to glitch!

```
1
2 module lock_fsm (input clk_in, rst_in, b0_in, b1_in, [4:0] code_in,
3                 output logic unlock_out);
4     parameter IDLE = 0;
5     parameter ENTRY = 1;
6
7     logic [3:0] entered_count;
8     logic state; //1 bits 2 states!
9     logic [4:0] vals_entered;
10
11    //State Transition and Transition Logic and Output Logic
12    always_ff @(posedge clk_in) begin
13        if (rst_in)begin
14            state <= IDLE;
15            vals_entered <= 5'b0;
16            entered_count <= 4'b0;
17        end else if (entered_count < 5 && (b0_in||b1_in))begin
18            state <= ENTRY;
19            vals_entered <= {vals_entered[3:0],b1_in};
20            entered_count <= entered_count + 1;
21        end
22        unlock_out <= (state==ENTRY)&&(entered_count==5)&&(vals_entered == code_in);
23    end
24 endmodule
```

Summary

- Other options (?):
 - Mealy (both of previous ones were basically MOORE state machines)
- No matter what many solutions will theoretically be a Mealy or Moore FSM...whether you structure it as such is up to you.
- Thinking about problems as Mealy/Moore FSM is a powerful way to get started on a solution
- Sometimes it can be too-restrictive and/or not scale the best, however
- The best choice is usually something in between!

Ray Tracer



Sam Gross, Adam Lerer - Spring 2007

Pong that you Live In



Nicholas Waltman Mike Wang, 2018